

# MATHEMATIK für die ASTRONOMIE

KULTURBUND DER DDR

Zentrale Kommission Astronomie und Raumfahrt – Arbeitskreis „Numerische Astronomie“

Veröffentlichung 5

Leipzig 1987

## Die Reduktion astronomischer Beobachtungen – das BASIC-Programm KORSA

Eine astronomische Positionsbestimmung ergibt sphärische Koordinaten in einem ortsveränderlichen Koordinatensystem.

Um solche Positionen vergleichen zu können, müssen sie in ein raumfestes Koordinatensystem transformiert werden.

Dieses Koordinatensystem sollte außerdem zeitunabhängig sein.

Zur Definition der astronomischen Koordinatensysteme wird jedoch der Frühlingspunkt verwendet, dessen Position sich zeitlich langsam ändert.

Es besteht also außerdem die Aufgabe, die Position in einem Koordinatensystem zu jedem interessierenden Zeitpunkt angeben zu können.

Das folgende Schema erläutert die Einzelheiten:

### für Mond und Planeten

beobachtbarer Ort

scheinbarer Ort,  
topozentrisch

scheinbarer Ort,  
geozentrisch

*wahrer*  
mittlerer Ort

*w. Ort*  
wahrer Ort zum  
Beobachtungszeitpunkt

*w. Ort*  
wahrer Ort zur  
Epoche

Refraktion

← tägl. Parallaxe jährl. →

tägliche und jährliche  
Aberration

Nutation

Präzession

### für Fixsterne

beobachtbarer Ort

scheinbarer Ort,  
geozentrisch

scheinbarer Ort,  
heliozentrisch

*wahrer*  
mittlerer Ort

*w. Ort*  
wahrer Ort zum  
Beobachtungszeitpunkt

*w. Ort*  
wahrer Ort zur  
Epoche

### Programmbeschreibung:

Mit dem Programm "KORSA" können alle Reduktionen und Koordinatentransformationen komfortabel ausgeführt werden.

Die Genauigkeit der verwendeten Verfahren ist besser als 0.1 Bogenminute – mit Ausnahme der Refraktion in Horizontnähe.

Diese Genauigkeit kann jedoch nur erreicht werden, wenn die Stellenzahl des verwendeten BASIC-Interpreters genügend groß ist (mind. acht Stellen).

- Eingabe und Speichern von Daten
- Koordinatentransformation
- Reduktion

Alle notwendigen Eingabedaten werden vom Programm im Dialog abgefragt. Durch ein System von Vorgabewerten wird der Eingabeaufwand verringert.

Im Programm selbst wurde mit der Unterprogrammtechnik gearbeitet, d.h. für alle Teilaufgaben stehen abgeschlossene Unterprogramme zur Verfügung. Die Eintrittspunkte und die zu übergebenden Variablen sind im Listing erläutert. Es ist also möglich, diese Unterprogramme einzeln in eigene Programme einzubauen oder auszutauschen.

#### Anpassung an verschiedene BASIC-Versionen

Das Programm wurde so geschrieben, daß eine Anpassung an andere BASIC-Dialekte ohne allzugroßen Aufwand möglich ist.

Die vorliegende Version läuft ungeändert auf Sinclair-Rechnern.

Für andere BASIC-Varianten müssen u.U. folgende Dinge geändert werden:

- Der Befehl "CLS" - Bildschirm löschen existiert häufig nicht. Er kann entweder weggelassen oder durch Ausgabe des entsprechenden Steuerzeichens (z.B. PRINT CHR\$(12)) ersetzt werden.
- Im verwendeten BASIC-Interpreter muß bei der Dimensionierung von String-Feldern die gemeinsame Länge aller Elemente angegeben werden. Bei vielen BASIC-Interpretern ist das nicht so; Zeile 1190 muß in DIM A\$(10,2) geändert werden. Die Unterprogramme 9900 und 9950 und ihre Aufrufe in den Zeilen 1740 und 1760 entfallen.
- Die mathematischen Funktionen "ASN" (arcsin) und "ACS" (arccos) sind meist nicht implementiert. Sie müssen durch die entsprechenden arithmetischen Ausdrücke

$$\arcsin(x) = \arctan\left(\frac{x}{\sqrt{1-x^2}}\right); \arccos(x) = \frac{\pi}{2} - \arcsin(x)$$

ersetzt werden.

- Der natürliche Logarithmus wird mitunter nicht durch "LN", sondern durch "LOG" programmiert.
- Bei Sinclair-Rechnern wird die Eingabe in einem besonderen Fenster realisiert. Die eingegebenen Werte stehen deshalb nicht automatisch auf dem Schirm. Wenn eine solche Fenstersteuerung nicht möglich ist, dann müssen die einem INPUT folgenden PRINT-Befehle (Echo) gestrichen werden.
- Die Konstante PI ist oft nicht vorhanden. Es ist eine Zeile 1001 LET PI=3.14159265358 nötig

Programmlisting "KORSA"

```
0002 REM "KORSA"
0100 PRINT "          KORSA"
0105 PRINT "          _____"
0106 PRINT
0107 PRINT
0110 PRINT
0120 PRINT "KOORDINATENTRANSFORMATION UND"
0130 PRINT "___"
0140 PRINT "REDUKTIONSRECHNUNG IN DER"
0150 PRINT "___"
0160 PRINT "SPHAERISCHEN ASTRONOMIE"
0170 PRINT "___"
0180 PRINT
0190 PRINT
0200 PRINT "VERSION 1.0 VOM APRIL 1987"
0210 PRINT
0220 PRINT "PROGRAMMIERER:"
0230 PRINT "    UWE PILZ"
0240 PRINT "    LAZARUSSTR.23"
0250 PRINT "    7024 LEIPZIG"
0300 INPUT B#
1000 REM INIT
1002 LET YE=1987
1003 LET MO=4
1004 LET TA=19
1010 LET I#="SPEICHER NR(1..10):"
1030 LET M#="AZIMUT"
1040 LET N#="HOEHE"
1050 LET O#="STUNDENWINKEL"
1060 LET P#="DEKLINATION"
1070 LET Q#="REKTASZENSION"
1080 LET R#="EKL. LAENGE"
1090 LET S#="EKL. BREITE"
1100 LET T#="EINGABE/SPEICHERN"
1110 LET U#="KOORDINATENTRANSFORMATION"
1120 LET V#="REDUKTION"
1130 LET W#="-----"
1135 LET X#=" AEQUATORSYSTEM"
1136 LET Y#=" ENTFERNEN"
1137 LET Z#=" HINZUFUEGEN"
1140 LET K1=0
1150 LET K2=0
1160 LET K#=""
1170 LET L#=""
1180 DIM A(10,2)
1190 DIM A#(10,2,13)
1200 REM EINGABE
1210 CLS
1220 PRINT T#
1230 PRINT W#
1240 PRINT "0 HORIZONTSYSTEM"
1250 PRINT "1 STUNDENWINKELSYSTEM"
1260 PRINT "2 AEQUATORSYSTEM"
1270 PRINT "3 EKLIPTIKSYSTEM"
1280 PRINT "4 DATEN NACH SPEICHER"
1290 PRINT "5 DATEN VOM SPEICHER"
1300 PRINT
1310 PRINT "ANDERE MENUES:"
1320 PRINT "6 ";U#
1330 PRINT "7 ";V#
1340 PRINT
```

```

1360 INPUT OP
1370 LET OP=INT(OP)
1380 IF (OP<0) OR (OP>7) THEN GOTO 1360
1410 IF OP>3 THEN GOTO 1540
1420 LET K $\alpha$ =M $\alpha$ 
1430 LET L $\alpha$ =N $\alpha$ 
1440 IF OP=0 THEN GOTO 1520
1450 LET K $\alpha$ =O $\alpha$ 
1460 LET L $\alpha$ =P $\alpha$ 
1470 IF OP=1 THEN GOTO 1520
1480 LET K $\alpha$ =Q $\alpha$ 
1490 IF OP=2 THEN GOTO 1520
1500 LET K $\alpha$ =R $\alpha$ 
1510 LET L $\alpha$ =S $\alpha$ 
1520 GOSUB 7500
1530 GOTO 1200
1540 IF OP<>4 THEN GOTO 1650
1543 CLS
1545 PRINT "DATEN NACH SPEICHER"
1550 PRINT "SPEICHER NR(1..10):";
1560 INPUT SP
1570 LET SP=INT(SP)
1580 IF (SP<1) OR (SP>10) THEN GOTO 1560
1590 PRINT SP
1600 LET A(SP,1)=K1
1610 LET A(SP,2)=K2
1620 LET A $\alpha$ (SP,1)=K $\alpha$ 
1630 LET A $\alpha$ (SP,2)=L $\alpha$ 
1640 GOTO 1770
1650 IF OP<>5 THEN GOTO 1800
1653 CLS
1655 PRINT "DATEN VOM SPEICHER"
1660 PRINT "SPEICHER NR(1..10):";
1670 INPUT SP
1680 LET SP=INT(SP)
1690 IF (SP<1) OR (SP>10) THEN GOTO 1720
1700 PRINT SP
1710 LET K1=A(SP,1)
1720 LET K2=A(SP,2)
1730 LET K $\alpha$ =A $\alpha$ (SP,1)
1740 GOSUB 9900
1750 LET L $\alpha$ =A $\alpha$ (SP,2)
1760 GOSUB 9950
1770 GOSUB 8000
1780 INPUT B $\alpha$ 
1790 GOTO 1200
1800 IF OP=6 THEN GOTO 2000
1810 GOTO 6300
2000 REM KOORDINATENTRANSFORMATION
2013 LET FD=0
2015 LET FZ=0
2017 LET FG=0
2019 CLS
2020 PRINT U $\alpha$ 
2030 PRINT W $\alpha$ 
2040 PRINT "0 HORIZONTSYSTEM"
2050 PRINT "1 STUNDENWINKELSYSTEM"
2060 PRINT "2 AEQUATORSYSTEM"
2070 PRINT "3 EKLIPTIKSYSTEM"
2080 PRINT
2090 PRINT "ANDERE MENUES:"

```

```

2100 PRINT "4 "; Tα
2110 PRINT "5 "; Vα
2120 PRINT
2130 INPUT OP
2140 LET OP=INT(OP)
2150 IF (OP<0) OR (OP>5) THEN GOTO 2130
2160 IF OP=4 THEN GOTO 1200
2170 IF OP=5 THEN GOTO 6300
2180 LET S0=4
2190 IF Kα=Mα THEN LET S0=1
2200 IF Kα=0α THEN LET S0=2
2210 IF Kα=Qα THEN LET S0=3
2220 LET S1=OP+1
2230 IF S0=S1 THEN GOTO 2019
2235 CLS
2240 IF S0>S1 THEN GOTO 2330
2250 FOR I=S0 TO S1-1
2260 IF I=1 THEN GOSUB 2400
2270 IF I=2 THEN GOSUB 3100
2280 IF I=3 THEN GOSUB 3500
2290 NEXT I
2300 GOSUB 8000
2310 INPUT Bα
2320 GOTO 2019
2330 FOR I=S0 TO S1+1 STEP-1
2340 IF I=4 THEN GOSUB 3700
2350 IF I=3 THEN GOSUB 3200
2360 IF I=2 THEN GOSUB 2600
2370 NEXT I
2380 GOTO 2300
2400 REM UP:HORIZONTS.-ST.WINK.S.
2410 REM INPUT UND OUTPUT:K1,K2
2420 REM FRAGT NACH GEOGR. KOORD.
2450 PRINT "HORIZONTSYSTEM-STUNDENWINKELSYS."
2460 GOSUB 3300
2490 LET K3=ASN(SIN(FI)*SIN(K2)-COS(FI)*COS(K2)*COS(K1))
2500 LET YY=SIN(K1)
2510 LET XX=COS(FI)*TAN(K2)+SIN(FI)*COS(K1)
2520 GOSUB 4280
2530 LET K1=WW
2540 LET K2=K3
2550 LET Kα=0α
2560 LET Lα=Pα
2570 RETURN
2600 REM UP:ST.WINK.S.-HORIZ.S.
2610 REM INPUT UND OUTPUT:K1,K2
2620 REM FRAGT NACH GEOGR. KOORD.
2640 PRINT "STUNDENWINKELSYSTEM-HORIZONTSYS."
2650 GOSUB 3300
2690 LET K3=ASN(SIN(FI)*SIN(K2)+COS(FI)*COS(K2)*COS(K1))
2700 LET YY=SIN(K1)
2710 LET XX=COS(K1)*SIN(FI)-TAN(K2)*COS(FI)
2720 GOSUB 4280
2725 LET K2=K3
2730 LET K1=WW
2740 LET Kα=Mα
2750 LET Lα=Nα
2760 RETURN
2800 REM UP:SCHEINBARE STERNZEIT
2810 REM MEEUS,KAP 7
2830 GOSUB 3400

```

```

2840 LET TT=JD-2415020
2845 IF FF<.5 THEN LET TT=TT-1
2850 LET TT=(TT+.5)/36525
2860 LET TH=.276919398+100.0021359*TT+.000001075*TT*TT
2870 LET TH=TH-INT(TH)
2880 LET TH=TH*2*PI
2900 LET WI=TH+1.002737908/12*PI*(HH+MI/60+SK/3600)
2910 GOSUB 8500
2920 LET TH=WI
2930 REM WANDELN MITTL.-SCHEINB. SZ
2940 GOSUB 5000
2950 GOSUB 3000
2960 LET TH=TH+DP*COS(SE)
2970 REM --> ORTSSTERNZEIT
2975 GOSUB 3300
2980 LET TH=TH-LB
2985 RETURN
3000 REM UP: SCHIEFE DER EKLIPTIK
3010 REM MEEUS, KAP. 18
3020 REM INPUT: JD OUTPUT: SE
3040 LET SE=23.452 294-.013 0125*TT-.000 001 64*TT*TT+.000 000 503
    *TT*TT*TT
3050 LET SE=SE/180*PI
3060 RETURN
3100 REM UP: ST.W.S.-AEQU.S
3110 REM INPUT UND OUTPUT: K1
3120 REM FRAGT NACH DATUM
3150 PRINT "STUNDENW.S.-AEQU.S. (WAHR. ORT)"
3160 GOSUB 2800
3170 LET K1=TH-K1
3180 LET K2=Q2
3190 RETURN
3200 REM AEQ.-ST.W.
3210 REM ERL. S. 3100
3230 PRINT "AEQU.S. (WAHR. ORT)-STUNDENW.S."
3240 GOSUB 2800
3250 LET K1=TH-K1
3260 LET K2=02
3270 RETURN
3300 REM UP: GEOGR. KOORD (FRAGT NUR 1 X )
3310 IF FG=1 THEN RETURN
3320 GOSUB 6200
3330 LET FG=1
3340 RETURN
3400 REM UP: DATUM+UHRZEIT (FRAGT NUR 1 X)
3410 IF FZ=1 THEN RETURN
3412 IF FD=0 THEN GOTO 3420
3414 GOSUB 8735
3416 GOTO 3425
3420 GOSUB 8700
3425 LET FZ=1
3427 LET FD=1
3430 RETURN
3450 REM UP: DATUM (FRAGT NUR 1 X)
3460 IF FD=1 THEN RETURN
3470 GOSUB 8900
3480 LET FD=1
3490 RETURN
3500 REM UP: AEQU.S.-EKLIPTIKS.
3510 REM INPUT UND OUTPUT: K1, K2
3520 REM FRAGT NACH ZEIT

```

```

3535 PRINT "AEQUATORSYSTEM-EKLIPTIKSYSTEM"
3540 GOSUB 3450
3550 GOSUB 3000
3560 LET K3=ASN(COS(SE)*SIN(K2)-SIN(SE)*COS(K2)*SIN(K1))
3570 LET YY=SIN(K1)*COS(SE)+TAN(K2)*SIN(SE)
3580 LET XX=COS(K1)
3582 GOSUB 4280
3584 LET K1=WW
3586 LET K2=K3
3590 LET Kα=Rα
3600 LET Lα=Sα
3610 RETURN
3700 REM UP:EKLIPTIKS.-AEQ.S.
3710 REM INPUT UND OUTPUT:K1,K2
3720 REM FRAGT NACH ZEIT
3740 PRINT "EKLIPTIKSYSTEM-AEQUATORSYSTEM"
3750 GOSUB 3450
3760 GOSUB 3000
3770 LET K3=ASN(COS(SE)*SIN(K2)+SIN(SE)*COS(K2)*SIN(K1))
3780 LET YY=SIN(K1)*COS(SE)-TAN(K2)*SIN(SE)
3790 LET XX=COS(K1)
3792 GOSUB 4280
3794 LET K1=WW
3796 LET K2=K3
3800 LET Kα=Qα
3810 LET Lα=Pα
3820 RETURN
4000 REM UP:PRAEZESSION
4010 REM INPUT:JO,FO,JD,FF,AO,DO
4020 REM OUTPUT:A1,D1
4040 REM JO,FO:AUSG.PERIODE
4050 REM JD,FF:ZIELPERIODE
4060 REM AO,DO:AUSG.KOORD.
4070 REM A1,D1:ZIELKOORD.
4080 REM BENUTZT:AA,BB,CC,FS,RR, TO,TH,TT,WW,XX,YY,ZT,ZZ
4110 LET FS=PI/180/3600
4120 LET TO=JO-2415020
4130 LET TO=(TO+FO+.313)/36524.2199
4140 LET TT=JD-JO
4150 LET TT=(TT+FF-FO)/36524.2199
4160 LET ZT=FS*((2304.25+1.396*TO)*TT+.302*TT*TT+.018*TT*TT*TT)
4170 LET ZZ=ZT+FS*(.791*TT*TT+.001*TT*TT*TT)
4180 LET TH=FS*((2004.682-.853*TO)*TT-.426*TT*TT-.042*TT*TT*TT)
4190 LET AA=COS(DO)*SIN(AO+ZT)
4200 LET BB=COS(TH)*COS(DO)*COS(AO+ZT)-SIN(TH)*SIN(DO)
4210 LET CC=SIN(TH)*COS(DO)*COS(AO+ZT)+COS(TH)*SIN(DO)
4220 LET YY=AA
4230 LET XX=BB
4240 GOSUB 4280
4250 LET A1=WW+ZZ
4260 LET D1=ASN(CC)
4270 RETURN
4280 REM UP:KARPOL
4290 REM INPUT XX,YY
4300 REM OUTPUT RR,WW
4310 LET RR=SQR(XX*XX+YY*YY)
4320 IF XX<>0 AND YY<>0 THEN GOTO 4350
4330 LET WW=SGN(YY)*PI/2+SGN(XX)*(SGN(XX)*PI/2-PI/2)
4340 RETURN
4350 LET WW=ATN(YY/XX)-(SGN(XX)-1)*SGN(YY)*PI/2
4360 RETURN

```

```

4400 REM UP:PRAEZESSION:DIALOG
4405 CLS
4407 PRINT "PRAEZESSION"
4410 PRINT "AUSGANGSEPOCHE:"
4420 GOSUB 8900
4430 LET JO=JD
4440 LET FO=FF
4450 PRINT "ZIELEPOCHE:"
4460 GOSUB 8900
4470 LET AO=K1
4480 LET DO=K2
4500 GOSUB 4000
4510 LET K1=A1
4520 LET K2=D1
4530 GOSUB 8000
4540 INPUT B#
4550 RETURN
5000 REM UP:NUTATION
5010 REM INPUT:JD,FF
5020 REM OUTPUT:DP,DE
5030 REM BENUTZT:TT,FA,LL,LS,MM,MS,OM
5050 LET TT=JD-2415020
5060 LET TT=(TT+FF)/36525
5070 LET FA=PI/180
5080 LET LL=FA*(279.6967+36000.7689*TT+.000303*TT*TT)
5090 LET LS=FA*(270.4342+481267.8831*TT-.001133*TT*TT)
5100 LET MM=FA*(358.4758+35999.0498*TT-.00015*TT*TT)
5110 LET MS=FA*(296.1046+477198.8491*TT+.009192*TT*TT)
5120 LET OM=FA*(259.1833-1934.1420*TT+.002078*TT*TT)
5130 LET DP=- (17.2327+.01737*TT)*SIN(OM)-(1.2729+.00013*TT)*SIN(LL
+LL)
5140 LET DP=DP+.2088*SIN(OM+OM)-.2037*SIN(LS+LS)
5150 LET DP=DP+(.1261-.00031*TT)*SIN(MM)+.0675*SIN(MS)
5160 LET DP=DP-(.0497-.00012*TT)*SIN(LL+LL+MM)
5170 LET DP=DP-.0342*SIN(LS+LS-OM)-.0261*SIN(LS+LS+MS)
5180 LET DP=DP+.0214*SIN(LL+LL-MM)-.0149*SIN(LL+LL-LS-LS+MS)
5190 LET DP=DP+.0124*SIN(LL+LL-OM)+.0114*SIN(LS+LS-MS)
5200 LET DP=DP*FA/3600
5210 LET DE=(9.21+.00091*TT)*COS(OM)+(.5522-.00029*TT)*COS(LL+LL)
5220 LET DE=DE-.0904*COS(OM+OM)+.0884*COS(LS+LS)
5230 LET DE=DE+.0216*COS(LL+LL+MM)+.0183*COS(LS+LS-OM)
5240 LET DE=DE+.0113*COS(LS+LS+MS)-.0093*COS(LL+LL-MM)-.0066*COS(L
L+LL-OM)
5250 LET DE=DE*FA/3600
5260 RETURN
5500 REM UP:DREHUNG
5510 REM INPUT:XX,YY,DW
5520 REM OUTPUT XX,YY
5530 REM DW=DREHWINKEL
5560 GOSUB 4280
5570 LET WW=WW+DW
5580 LET XX=RR*COS(WW)
5590 LET YY=RR*SIN(WW)
5595 RETURN
5600 REM UP:NUTATION,DIALOG
5605 CLS
5610 PRINT "NUTATION";
5620 IF RI=-1 THEN PRINT Z#
5625 IF RI=1 THEN PRINT Y#
5630 GOSUB 8900
5633 GOSUB 5000

```



```

5635 GOSUB 3000
5640 LET DA=DP*(COS(SE)+SIN(SE)*SIN(K1)*TAN(K2))-DE*(COS(K1)*TAN(K
2))
5645 LET DD=DP*(SIN(SE)*COS(K1))+DE*SIN(K1)
5650 LET K1=K1-RI*DA
5655 LET K2=K2-RI*DD
5660 GOSUB 8000
5665 INPUT Bα
5670 RETURN
5700 REM UP: DATUM
5710 REM TA, MO, YE IN JD, FF
5720 REM JD, FF IN TA, MO, YE
5730 REM JULIANISCHES <-> BUERGERLICHES DATUM
5750 REM BUERG -> JD
5760 GOTO 5800
5770 REM JD -> BUERG.
5780 GOTO 5930
5800 LET D1=INT(TA)
5810 LET FF=TA-D1-.5
5820 LET JD=-INT(7*(INT((MO+9)/12)+YE)/4)
5830 LET SS=SGN(MO-9)
5840 LET AA=ABS(MO-9)
5850 LET J1=INT(YE+SS*INT(AA/7))
5860 LET J1=-INT((INT(J1/100)+1)*3/4)
5870 LET JD=JD+INT(275*MO/9)+D1+J1
5880 LET JD=JD+1721027+2+367*YE
5890 IF FF>=0 THEN GOTO 5940
5900 LET FF=FF+1
5910 LET JD=JD-1
5912 LET TT=JD-2415020
5914 LET TT=(TT+FF)/36525
5920 RETURN
5930 REM ANSPRUNG JD -> GREG.
5940 LET J1=JD
5950 LET F1=FF+.5
5960 IF F1<1 THEN GOTO 5990
5970 LET F1=F1-1
5980 LET J1=J1+1
5990 LET A1=INT((J1/36524.25)-51.12264)
6000 LET AA=J1+1+A1-INT(A1/4)
6010 LET BB=AA+1524
6020 LET CC=INT((BB/365.25)-.3343)
6030 LET DD=INT(365.25*CC)
6040 LET EE=INT((BB-DD)/30.61)
6050 LET TA=BB-DD-INT(30.61*EE)+F1
6060 LET MO=EE-1
6070 LET YE=CC-4716
6080 IF EE>13.5 THEN LET MO=MO-12
6090 IF MO<2.5 THEN LET YE=YE+1
6100 RETURN
6200 REM UP: GEOGR. KOORDINATEN FI UND LB ABFRAGEN
6210 PRINT "GEOGR. LAENGE (GRAD, MIN, SEC; 0:-) LAMBDA>180: DEFAULT
-13.25/52.5"
6212 LET LB=-13.25/180*PI
6214 LET FI=52.5/180*PI
6216 INPUT Bα
6218 IF VAL(Bα)>180 THEN RETURN
6220 GOSUB 7835
6225 IF WI>PI THEN RETURN
6230 LET LB=WI
6240 PRINT "GEOGR. BREITE (GRAD, MIN, SEC)"

```

2 x 30.0061

```

6250 GOSUB 7800
6260 LET FI=WI
6270 RETURN
6300 REM REDUKTION
6340 CLS
6350 PRINT V $\alpha$ 
6352 PRINT " ( -:";Y $\alpha$ ;" +: ";Z $\alpha$ ;" )"
6360 PRINT W $\alpha$ 
6370 PRINT "1 REFRAKTION      HORIZONTSYSTEM"
6380 PRINT
6390 PRINT "  PARALLAXE"
6400 PRINT "2 TAEGLICHE      HORIZONTSYSTEM"
6410 PRINT "3 JAEHRLICHE " EKLIPTIKSYSTEM"
6415 PRINT
6420 PRINT "  ABERRATION"
6430 PRINT "4 TAEGLICHE      STUNDENWINKELS."
6440 PRINT "5 JAEHRLICHE      EKLIPTIKSYSTEM"
6445 PRINT
6450 PRINT "6 NUTATION        ";X $\alpha$ 
6455 PRINT
6460 PRINT "7 PRAEZSSION     ";X $\alpha$ 
6461 PRINT
6462 PRINT "8 EIGENBEWEGUNG";X $\alpha$ 
6464 PRINT
6466 PRINT "  ANDERE MENUES"
6467 PRINT "9 ";T $\alpha$ 
6468 PRINT "0 ";U $\alpha$ 
6470 INPUT OP
6472 LET RI=-SGN(OP)
6480 LET OP=INT(ABS((OP)))
6490 IF OP<0 OR OP>9 THEN GOTO 6470
6492 IF OP=9 THEN GOTO 1200
6494 IF OP=0 THEN GOTO 2000
6500 IF OP<>1 THEN GOTO 6560
6510 IF K $\alpha$ =M $\alpha$  THEN GOTO 6540
6515 CLS
6520 PRINT "FALSCHES KOORDINATENSYSTEM."
6524 INPUT B $\alpha$ 
6530 GOTO 2000
6540 GOSUB 7200
6550 GOTO 6340
6560 IF OP<>2 THEN GOTO 6600
6570 IF K $\alpha$ <>M $\alpha$  THEN GOTO 6515
6580 GOSUB 9100
6590 GOTO 6340
6600 IF OP<>3 THEN GOTO 6640
6610 IF K $\alpha$ <>R $\alpha$  THEN GOTO 6515
6620 GOSUB 9200
6630 GOTO 6340
6640 IF OP<>4 THEN GOTO 6680
6650 IF K $\alpha$ <>O $\alpha$  THEN GOTO 6515
6660 GOSUB 9300
6670 GOTO 6340
6680 IF OP<>5 THEN GOTO 6720
6690 IF K $\alpha$ <>R $\alpha$  THEN GOTO 6515
6700 GOSUB 9400
6710 GOTO 6340
6720 IF OP<>6 THEN GOTO 6760
6730 IF K $\alpha$ <>Q $\alpha$  THEN GOTO 6515
6740 GOSUB 5600
6750 GOTO 6340

```

```

6760 IF OP<>7 THEN GOTO 6800
6770 IF K<>Q THEN GOTO 6515
6780 GOSUB 4400
6790 GOTO 6340
6800 REM EIGENBEW.
6802 CLS
6804 PRINT "EIGENBEWEGUNG"
6810 PRINT "AUSGANGSZEITPUNKT"
6820 GOSUB 8900
6830 LET J2=JD
6840 PRINT "ZIELZEITPUNKT"
6850 GOSUB 8900
6870 LET DJ=(JD-J2)/36525
6880 PRINT "100-JAHRIGE EIGENBEWEGUNG"
6890 PRINT "REKTASZENSION"
6900 PRINT "ZEIT- ODER WINKELSEKUNDEN (Z/S)";
6910 INPUT B
6920 IF B<>"Z" AND B<>"S" THEN GOTO 6910
6930 PRINT B
6940 LET FZ=1
6945 IF B="Z" THEN LET FZ=15
6950 INPUT DR
6955 PRINT DR;" SEC/JH"
6960 PRINT "DEKLINATION - WINKELSEKUNDEN"
6965 INPUT DD
6970 PRINT DD;" SEC/JH"
6975 LET K1=K1+DJ*DR/3600/180*PI*FZ
6980 LET K2=K2+DJ*DD/3600/180*PI
6985 GOSUB 8000
6990 INPUT B
6995 GOTO 6340
6999 STOP
7000 REM UP:REFRAKTION
7010 REM INPUT:ZO, PL, TL, LD
7020 REM OUTPUT:RE
7030 REM ZO:ZENITDISTANZ - RAD
7040 REM PL:LUFTDRUCK - PASCAL
7050 REM TL:LUFTTEMPERATUR - KELVIN
7060 REM LD:LICHTWELLENLAENGE - NANOMETER
7070 REM RE:REFRAKTION - RAD
7090 LET R0=-2.64678*LN(LD)+77.1559
7100 LET R1=-.00260936*LN(LD)+.083506
7110 LET TZ=TAN(ZO)
7120 LET RE=(R0*TZ-R1*TZ*TZ*TZ)*PL/101323.2*273/TL
7130 LET RE=RE/3600*PI/180
7140 RETURN
7200 REM UP:REFRAKTION:DIALOG
7210 CLS
7220 PRINT "REFRAKTION";
7222 IF RI=-1 THEN PRINT Z
7224 IF RI=1 THEN PRINT Y
7230 PRINT "STANDARDWERTE (INPUT 0):"
7240 PRINT "T=0 GRAD C, P=1013 MBAR, LA=580 NM"
7250 PRINT "TEMPERATUR(GRAD C):";
7260 INPUT TL
7270 PRINT TL
7280 LET TL=TL+273
7290 PRINT "LUFTDRUCK(MBAR):";
7300 INPUT PL
7310 IF PL=0 THEN LET PL=1013.232
7320 PRINT PL

```

```

7330 LET PL=PL*100
7340 PRINT "WELLENLAENGE (NM):";
7350 INPUT LD
7360 IF LD=0 THEN LET LD=580
7370 PRINT LD
7380 LET Z0=PI/2-K2
7385 GOSUB 7000
7390 PRINT "REFRAKTION=";RE/PI*180*3600;" SEC."
7410 LET K2=K2-RI*RE
7420 GOSUB 8000
7430 INPUT B#
7440 RETURN
7500 REM UP:K1,K2 ABFRAGEN UND IN RAD WANDELN
7510 REM INPUT:K#,L#
7520 REM OUTPUT:K1,K2
7540 IF (K#<>Q#) AND (K#<>0#) THEN GOTO 7710
7550 PRINT K#;"(ST,MIN,SEC):"
7560 INPUT AL
7570 IF AL<0 THEN LET AL=AL+24
7580 IF AL<0 THEN GOTO 7570
7590 IF AL<10 THEN PRINT " ";
7600 PRINT AL;" ";
7610 INPUT AM
7620 IF AM<0 THEN GOTO 7610
7630 IF AM<10 THEN PRINT "0";
7640 PRINT AM;" ";
7650 INPUT AS
7660 IF AS<0 THEN GOTO 7650
7670 IF AS<10 THEN PRINT "0";
7680 PRINT AS
7690 LET K1=PI/12*(AL+AM/60+AS/3600)
7700 GOTO 7740
7710 PRINT K#;"(GRAD,MIN,SEC):"
7720 GOSUB 7800
7730 LET K1=WI
7740 PRINT L#;"(GRAD,MIN,SEC):"
7750 GOSUB 7800
7760 LET K2=WI
7770 RETURN
7800 REM UP:WINKEL WI ABFRAGEN UND WANDELN IN RAD
7810 REM OUTPUT:WI
7830 INPUT B#
7835 REM EINTRITT B# SCHON BEKANNT (DEFAULTBEH.)
7836 LET SW=SGN(VAL(B#+ "1"))
7837 LET WG=ABS(VAL(B#))
7838 IF SW=-1 THEN PRINT "-";
7840 PRINT WG;" GRAD ";
7850 INPUT WM
7860 IF (WM<0) OR (WM>=60) THEN GOTO 7850
7870 PRINT WM;" MIN ";
7880 INPUT WS
7890 IF (WS<0) OR (WS>=60) THEN GOTO 7870
7900 PRINT WS;" SEC"
7930 LET WI=SW*PI/180*(WG+WM/60+WS/3600)
7940 RETURN
8000 REM UP:K1,K2 AUSGEBEN
8010 REM INPUT:K1,K2,K#,L#
8030 LET WI=K2
8040 GOSUB 8500
8050 LET K2=WI
8060 LET WI=K1

```

```

8070 GOSUB 8600
8080 LET K1=WI
8090 IF (K1<>Q1) AND (K1<>0) THEN GOTO 8240
8100 PRINT K1;"(ST,MIN,SEC)"
8110 LET WI=WI*12/PI
8120 IF WI<0 THEN LET WI=WI+24
8130 LET WH=INT(WI)
8140 LET WI=60*(WI-WH)
8150 LET WM=INT(WI)
8160 LET WS=60*(WI-WM)
8170 IF WH<10 THEN PRINT " ";
8180 PRINT WH;":";
8190 IF WM<10 THEN PRINT "0";
8200 PRINT WM;":";
8210 IF WS<10 THEN PRINT "0";
8220 PRINT WS
8230 GOTO 8260
8240 PRINT K1;":"
8250 GOSUB 8300
8260 PRINT L1;":"
8270 LET WI=K2
8280 GOSUB 8300
8290 RETURN
8300 REM UP:WINKEL IN GRAD,MIN,SEC WANDELN UND AUSGEBEN
8310 REM INPUT:WI
8330 LET SW=SGN(WI)
8340 LET WI=ABS(WI*180/PI)
8350 LET WG=INT(WI)
8360 LET WI=60*(WI-WG)
8370 LET WM=INT(WI)
8380 LET WS=60*(WI-WM)
8390 IF SW=-1 THEN PRINT "-";
8400 PRINT WG;" GRAD ";WM;" MIN ";WS;" SEC"
8410 RETURN
8500 REM UP:WINKEL WI AUF -PI<WI<=PI REDUZIEREN
8520 LET SG=SGN(WI)
8530 LET WI=ABS(WI)
8540 LET WI=2*PI*(WI/2/PI-INT(WI/2/PI))
8550 LET WI=SG*WI
8560 RETURN
8600 REM UP:WINKEL WI AUF 0<=WI<2*PI REDUZIEREN
8620 GOSUB 8500
8630 IF WI<0 THEN LET WI=WI+PI+PI
8640 RETURN
8700 REM UP:DATUM+UHRZEIT-->J.D.
8710 REM OUTPUT:JD,FF
8730 GOSUB 8900
8735 REM EINTITT:UHRZEIT=HH,MI,SK
8737 PRINT "UNIVERSAL TIME (ENTER=LETZTE)"
8740 PRINT "STUNDE:";
8750 INPUT B1
8751 IF B1<>" " THEN LET HH=VAL(B1)
8752 LET HH=INT(HH)
8754 IF (HH<0) OR (HH>=24) THEN GOTO 8750
8760 PRINT HH;" MINUTE:";
8770 IF B1<>" " THEN INPUT MI
8782 LET MI=INT(MI)
8784 IF (MI<0) OR (MI>=60) THEN GOTO 8780 8770
8790 PRINT MI;" SEK:";
8810 IF B1<>" " THEN INPUT SK
8812 IF (SK<0) OR (SK>=60) THEN GOTO 8810

```

```

8820 PRINT SK
8825 LET TS=TA
8830 LET TA=TA+HH/24+MI/1440+SK/86400
8835 LET TA=TS
8840 GOSUB 5760
8850 RETURN
8900 REM UP:BUERG. DATUM ABFRAGEN UND IN J.D. WANDELN
8910 REM OUTPUT:JD,FF
8924 PRINT "DATUM:"
8925 PRINT "(ENTER:LETZTES, MONAT=0:TROP.J.)"
8930 PRINT "JAHR:";
8935 INPUT B#
8936 IF B#<>" " THEN LET YE=VAL(B#)
8950 PRINT YE;" MONAT:";
8970 IF B#<>" " THEN INPUT MO
8971 IF MO<0 OR MO>12 THEN GOTO 8970
8972 IF MO<>0 THEN GOTO 8989
8973 LET JD=(YE-1900)*365.242199
8974 LET FF=JD-INT(JD)+.313
8975 IF FF<1 THEN GOTO 8978
8976 LET FF=FF-1
8977 LET JD=JD+1
8978 LET JD=INT(JD)+2415020
8979 GOSUB 5770
8980 LET HH=24*(TA-INT(TA))
8981 LET TA=INT(TA)
8982 LET MI=60*(HH-INT(HH))
8983 LET HH=INT(HH)
8984 LET SK=60*(MI-INT(MI))
8985 LET MI=INT(MI)
8986 LET B#=""
8987 PRINT
8988 PRINT "JAHR:";YE;" MONAT:";
8989 PRINT MO;" ";
8990 PRINT "TAG:";
9000 IF B#<>" " THEN INPUT TA
9010 PRINT TA
9020 GOSUB 5760
9030 RETURN
9100 REM UP:TAEGL.PARRAL. DIALOG
9105 REM INPUT UND OUTPUT:K1,K2
9107 CLS
9108 PRINT "TAEGL.PARALLAXE";
9110 IF RI=-1 THEN PRINT Z#
9115 IF RI=1 THEN PRINT Y#
9120 PRINT "ENTFERNUNG/PARALLAXENWINKEL E/P?"
9125 INPUT B#
9130 IF B#<>"E" AND B#<>"P" THEN GOTO 9125
9135 IF B#="E" THEN PRINT "ENTFERNUNG (AE):";
9140 IF B#="P" THEN PRINT "PARALLAXENWINKEL (SEC):";
9145 INPUT PW
9150 PRINT PW
9155 IF B#="E" THEN LET PW=ASN(6360/149.6E6/PW)
9160 IF B#="P" THEN LET PW=PW/3600/180*PI
9165 LET PH=PW*COS(K2)
9170 PRINT "PARALLAXE=";PH/PI*180*3600;" SEC"
9180 LET K2=K2+RI*PH
9186 GOSUB 8000
9190 INPUT B#
9195 RETURN
9200 REM UP:JAEHRL. PARALLAXE,DIALOG

```

```

9203 REM INPUT UND OUTPUT:K1,K2
9205 CLS
9210 PRINT "JAEHRL. PARALLAXE";
9215 IF RI=-1 THEN PRINT Z□
9220 IF RI=1 THEN PRINT Y□
9235 GOSUB 9800
9240 PRINT "ENTFERNUNG (LJ)=";
9245 INPUT LJ
9250 PRINT LJ
9255 LET PW=149.6E6/9.461E12/LJ
9260 LET DL=PW/COS(K2)*SIN(K1-LS)
9265 LET DB=PW*SIN(K2)*COS(K1-LS)
9270 LET K1=K1+RI*DL
9275 LET K2=K2+RI*DB
9280 GOSUB 8000
9285 INPUT B□
9290 RETURN
9300 REM UP: TA EGL. ABERRATION, DIALOG
9302 REM INPUT UND OUTPUT:K1,K2
9305 CLS
9310 PRINT "TA EGL. ABERRATION ";
9315 IF RI=-1 THEN PRINT Z□
9320 IF RI=1 THEN PRINT Y□
9325 PRINT "GEOG. BREITE (>90=DEFAULT 52.5):"
9330 INPUT WG
9335 IF WG>90 THEN LET WI=52.5
9337 IF WG>90 THEN GOTO 9343
9340 GOSUB 7835
9343 LET AK=0.32/3600/180*PI*COS(WI)
9350 LET K1=K1-RI*AK/COS(K2)*COS(K1)
9355 LET K2=K2-RI*AK*SIN(K2)*SIN(K1)
9360 GOSUB 8000
9365 INPUT B□
9370 RETURN
9400 REM UP: JAEHRL. ABERRATION MIT DIALOG
9402 REM INPUT UND OUTPUT:K1,K2
9403 CLS
9405 PRINT "JAEHRL. ABERRATION";
9410 IF RI=-1 THEN PRINT Z□
9415 IF RI=1 THEN PRINT Y□
9420 GOSUB 9800
9425 LET AK=20.47/3600/180*PI
9430 LET DL=-AK/COS(K2)*COS(K1-LS)
9435 LET DB=AK*SIN(K2)*SIN(K1-LS)
9440 LET K1=K1-RI*DL
9445 LET K2=K2-RI*DB
9450 GOSUB 8000
9455 INPUT B□
9460 RETURN
9800 REM UP: LS=WAHRE SONNENLAENGE, FRAGT NACH DATUM
9810 GOSUB 8700
9815 LET TT=JD-2415020
9820 LET TT=(TT+FF)/36525
9825 LET LL=279.69668+36000.76892*TT+.0003025*TT*TT
9835 LET MM=358.47583+35999.04975*TT-.00015*TT*TT-.0000033*TT*TT*T
9840 LET WI=MM/180*PI
9850 GOSUB 8500
9855 LET MM=WI
9860 LET CC=(1.91946-.004789*TT-.000014*TT*TT)*SIN(MM)+( .020094-.0
001*TT)*SIN(MM+MM)+.000293*SIN(3*MM)
9865 LET WI=PI/180*(LL+CC)

```

```
9870 GOSUB 8500
9875 LET LS=WI
9880 RETURN
9900 REM UP:K $\alpha$  KUERZEN (NUR SINCLAIR) AUFRUF:1740
9910 IF K $\alpha$ =" " OR K $\alpha$ (LEN(K $\alpha$ ))<>" " THEN RETURN
9920 LET K $\alpha$ =K $\alpha$ ( TO LENK $\alpha$ -1)
9930 GOTO 9910
9950 REM UP:L $\alpha$  KUERZEN (NUR SINCLAIR) AUFRUF:1760
9960 IF L $\alpha$ =" " OR L $\alpha$ (LENL $\alpha$ )<>" " THEN RETURN
9970 LET L $\alpha$ =L $\alpha$ ( TO LENL $\alpha$ -1)
9980 GOTO 9960
```

Anschrift des Autors

Uwe Pilz  
Lazarusstr. 23  
Leipzig  
7024

Der Autor ist Mitglied im

Arbeitskreis "Numerische Astronomie"  
(Leiter: Dipl.-Phys. L. Ehrenberg)  
PSF 29  
Schkeuditz  
7144